
RTC M41T56C64:
dokładny chronometr 
z pamięcią EEPROM

Układ M41T56C64, produkowa-
ny przez firmę STMicroelectronics 
(www.st.com) jest niebanalnym po-
łączeniem układu RTC M41T56, 
pamięci EEPROM M24C64 oraz 
rezonatora kwarcowego 32768 Hz 
w 18–wyprowadzeniowej obudo-
wie do montażu powierzchniowe-
go. Rozmieszczenie wyprowadzeń 
układu przedstawiono na rys. 1. 
Aż 9 wyprowadzeń układu (NC, 
NF i VSS) przeznaczonych jest do 
połączenia z masą układu, dwa 
służą do dołączenia napięcia za-
silającego (VCC) i baterii (VBAT). Po-
zostaje więc jedynie 7 wyprowa-
dzeń funkcyjnych. Wyprowadzenia 
E0...E2 służą do określenia trzech 
młodszych bitów adresu układu 
pamięci EEPROM na magistrali I2C 
– ich przeznaczenie jest identycz-
ne jak w „samodzielnym” układzie 
M24C64. Również wyprowadzenie 
/WC jest powiązane z pamięcią 
EEPROM i również jego funkcja 
jest identyczna jak w układzie 
M24C64. Wyprowadzenie FT/OUT 
pełni podwójną funkcję. Pracując 
w trybie FT (frequency test) sta-
nowi wyjście sygnału zegarowego 
podzielonego przez 64, czyli na 
wyjściu tym generowany jest syg-
nał o częstotliwości 512 Hz, który 

Pomiar czasu jest jednym z podstawowych pomiarów wykonywanych 
we współczesnych urządzeniach elektronicznych. Zegary mamy w te-
lefonach komórkowych, aparatach fotograficznych, mikrofalówkach, 
odtwarzaczach MP3 itd., itd. W większości powyższych przypadków 
super dokładność pomiaru czasu nie jest bardzo istotna, ale mimo 
tego chcielibyśmy, aby punktualność naszego ze-
gara była jak największa, choćby ze względu na 
unikanie konieczności dokonywania korekty wska-
zań. Dobrym rozwiązaniem jest więc stosowanie 
specjalizowanych układów.

Rys. 1. Rozmieszczenie wyprowadzeń 
układu M41T56C64 Rys. 2. Schemat blokowy układy M41T56C64

(1) NC – Wyprowadzenia powinny być połączone 
z masą układu.
(2) NF – Wyprowadzenia powinny być połączone 
z masą układu, wyprowadzenia 2 i 3 oraz 16 
i 17 są wewnętrznie połączone ze sobą
(3) Wyjście z otwartym drenem.

można wykorzystać do kalibracji 
częstotliwości wbudowanego oscy-
latora, natomiast w trybie OUT 
wyprowadzenie to pełni funkcję 
uniwersalnego wyjścia (z otwartym 
drenem), którego stan określa je-
den z bitów rejestrów układu ze-
gara. Schemat blokowy układu 
M41T56C64 przedstawiono na 
rys. 2. Z punktu widzenia progra-
misty piszącego program współpra-
cujący z układem M41T56C64 skła-
da się on z dwóch niezależnych 
układów: M41T56 oraz M24C64 
pracujących na wspólnej magistra-
li I2C. Obydwa bloki funkcjonalne 
zostaną omówione w dalszej części 
artykułu. Procedury i przykładowe 
programy ilustrujące artykuł zosta-
ły napisane w asemblerze dla mi-

krokontrolera ST7FLITE19 i mogą 
być uruchomione na komputerku 
LITEcomp. Schemat podłączenia 
układu M41T56C64 do LITEcompa 
przedstawiony jest na rys. 3. 

Procedury obsługi magistrali 
I2C

Procedury obsługi magistrali I2C 
zostały omówione w ramach kursu 
„LITEcomp – aplikacje” w numerze 
6/2007 Elektroniki Praktycznej. Dla 
przypomnienia publikujemy je na 
list. 1.

Zegar M41T56
C z ę ś ć  z e g a r o w a  u k ł a d u 

M41T56C64 jest funkcjonalnym 
odpowiednikiem samodzielnego 
układu M41T56. Posiada on 8 re-

List. 1. Procedury obsługi magistrali I2C
;***************************************************************************
; Procedura generująca sygnał START
;***************************************************************************
.I2C_Start
 BRES I2C_DIR, #SDA
 BRES I2C_DIR, #SCL
 BSET I2C_DIR, #SDA   ; wyzerowanie linii SDA
 CALL i2c_delay
 BSET I2C_DIR, #SCL   ; wyzerowanie linii SCL
 RET  

;***************************************************************************
; Procedura generująca sygnał STOP
;***************************************************************************
.I2C_Stop
 BRES I2C_DIR, #SCL   ; ustawienie linii SCL
 CALL i2c_delay
 BRES I2C_DIR, #SDA   ; ustawienie linii SDA
 RET

;***************************************************************************
; Procedura zapisująca bajt na magistralę I2C 
; wartość bitu ACK zwracana jest we wskaźniku C
; Wysyłany bajt należy umieścić w rejestrze A
;***************************************************************************
.I2C_Write
 LD i2c_data, A     ; zapamiętanie A w zmiennej i2c_data
 LD A, #8 
 LD i2c_counter, A    ; ilość powtórzeń pętli
i2cw1  
 BSET I2C_DIR, #SCL   ; stan niski na linii SCL
 BSET I2C_DIR, #SDA   ; stan niski na linii SDA
 LD A, i2c_data    ; przywrócenie danej
 RLC A       ; obrót w lewo przez C
 LD i2c_data, A    ; zapamiętanie danej
 JRNC i2cw2      ; skok, jeśli C wyzerowany
 BRES I2C_DIR, #SDA   ; w przeciwnym razie ustawienie linii SDA
i2cw2
 BRES I2C_DIR, #SCL   ; ustawienie linii SCL
 DEC i2c_counter   ; zmniejszenie licznika pętli
 JRNE i2cw1      ; skok do i2cw1 jeśli większe od zera
 BSET I2C_DIR, #SCL   ; wyzerowanie linii SCL
 BRES I2C_DIR, #SDA   ; wysoki stan na linii SDA
 BRES I2C_DIR, #SCL   ; wysoki stan na linii SCL
 BTJT I2C_DAT, #SDA, i2cw3 ; stan linii SDA do C
i2cw3
 BSET I2C_DIR, #SDA   ; wyzerowanie linii SDA
 BSET I2C_DIR, #SCL   ; wyzerowanie linii SCL
 RET

;***************************************************************************
; Procedura odczytująca bajt z magistrali I2C
; Bit ACK przed wywołaniem procedury należy umieścić we wskaźniku C
;***************************************************************************
.I2C_Read
 RLC A       ; zapamiętanie bitu ACK przekazanego poprzez C
 LD i2c_data, A
 LD A,#8
 LD  i2c_counter, A  ; ilość powtórzeń pętli
 BRES I2C_DIR, #SDA   ; stan wysoki na linii SDA
i2cr1
 CALL  i2c_delay    ; czekamy chwilkę
 BRES I2C_DIR, #SCL   ; stan wysoki na linii SCL
 BTJT I2C_DAT, #SDA, i2cr2 ; stan linii SDA do C
i2cr2
 LD A, i2c_data    ; przywracamy poprzednio odczytane bity
 RLC  A       ; zapamiętujemy nowo odebrany bit
 LD i2c_data, A    ; zapamiętujemy daną
 BSET I2C_DIR, #SCL   ; niski stan na linii SCL
 DEC i2c_counter   ; zmniejszamy licznik pętli
 JRNE i2cr1      ; skok do i2cr1 jeśli większe od zera
 BSET I2C_DIR, #SDA   ; zerujemy linię SDA
 JRNC i2cr3      ; skok do i2cr3 jeśli wskaźnik C wyzerowany
 BRES I2C_DIR, #SDA   ; ustawiamy linię SDA
i2cr3
 BRES I2C_DIR, #SCL   ; ustawiamy linię SCL
 BSET I2C_DIR, #SCL   ; zerujemy linię SCL
 BSET I2C_DIR, #SDA   ; zerujemy linię SDA
 LD A, i2c_data    ; w A odebrana dana
 RET

jestrów sterujących służących do 
ustawiania oraz odczytu informa-
cji o czasie i dacie oraz 56 bajtów 
pamięci RAM ogólnego przezna-
czenia. Strukturę rejestrów układu 
zegara przedstawiono na rys. 4. 
Informacja o czasie i dacie jest za-
pisana w formacie BCD. Struktu-
rę bajtu adresu układu M41T56 
przedstawiono na rys. 5. Adres 
ten jest ustawiony na stałe i nie 
ma możliwości jego zmiany, tak 
więc na jednej magistrali I2C 
może pracować tylko jeden układ 
M41T56/M41T56C64.

Zapis danych do rejestrów 
zegara

Przebiegi na magistrali I2C pod-
czas zapisu danych do rejestrów 
wewnętrznych zegara oraz pamię-
ci RAM przedstawiono na rys. 6. 
Możliwy jest sekwencyjny zapis 
kolejnych danych, gdyż adres re-
jestru wewnętrznego jest automa-
tycznie inkrementowany po każdej 
operacji zapisu. Kod procedury 
zapisu danej do rejestru układu 
M41T56 przedstawiono na list. 2.

Odczyt danych z rejestrów 
zegara

Przebiegi na magistrali I2C pod-
czas odczytu danych z rejestrów 
wewnętrznych zegara oraz pamię-
ci RAM przedstawiono na rys. 7. 
Podobnie jak w przypadku zapisu, 
adres rejestru wewnętrznego jest 
automatycznie inkrementowany po 
każdej operacji odczytu, przez co 
możliwy jest sekwencyjny odczyt 
danych. Procedurę odczytu danych 
z rejestru układu M41T56 przed-
stawiono na list. 3. 

Pamięć EEPROM
Pa m i ę ć  E E P R O M  z a w a r t a 

w układzie M41T56C64 jest funk-
cjonalnym odpowiednikiem ukła-
du pamięci M24C64 zawierające-
go 8192 bajty pamięci EEPROM. 
Strukturę bajtu adresu układu 
pamięci EEPROM przedstawiono 
na rys. 8. Stan bitów E2...E0 jest 
porównywany ze stanem wypro-
wadzeń E2...E0, dzięki czemu na 
magistrali I2C może pracować do 
ośmiu układów pamięci EEPROM, 
np. jeden układ M41T56C64 oraz 
do siedmiu klasycznych układów 
M24C64 lub innych układów pra-
cujących na magistrali I2C, któ-
rych starsza połówka bajtu adresu 
jest równa 1010.

Zapis danych do pamięci 
EEPROM

Zapis do komórki pamięci jest 
możliwy na dwa sposoby: za-
pis pojedynczej komórki pamięci 
oraz zapis strony o rozmiarze do 
32 bajtów. Przebiegi na magistrali 
podczas zapisu pojedynczego baj-
tu do pamięci przedstawiono na 
rys. 9. Transmisja jest rozpoczy-

nana od nadania przez układ ma-
ster sekwencji START oraz adresu 
układu SLAVE, który jest potwier-
dzony przez układ pamięci bitem 
ACK. Następnie transmitowane są 
dwa bajty składające się na adres 
komórki pamięci. Każdy z tych baj-
tów jest potwierdzany przez układ 
pamięci bitem ACK. Jako ostatni 
transmitowany jest bajt przezna-

Rys. 3. Schemat dołączenia układu M41T56C64 do LITECompa

Rys. 4. Struktura rejestrów zegara

Rys. 5. Struktura bajtu adresu zegara

Rys. 6. Przebiegi na magistrali I2C podczas zapisu danych do rejestrów zegara

czony do zapisania w pamięci. 
W zależności od stanu wyprowa-
dzenia /WC bajt ten jest potwier-
dzany bitem ACK (wyprowadzenie 
/WC w stanie niskim), bądź też 
układ pamięci transmituje bit NO-
ACK świadczący o niepomyślnym 
zakończeniu operacji zapisania 
bajtu do pamięci (wyprowadzenie 
/WC w stanie wysokim). Procedurę 
zapisu pojedynczego bajtu danych 
do pamięci EEPROM przedstawio-
ny jest na list. 4. 

Przed wywołaniem procedury 
należy do zmiennej m24DstAddr 
załadować adres, pod który ma 
zostać zapisana dana, natomiast 
w rejestrze A należy umieścić za-
pisywaną daną.

Przebiegi na magistrali pod-
czas zapisu strony 32 bajtów 
przedstawiono na rys. 10. Trans-
misja przebiega podobnie, jak 
w przypadku zapisu jednego bajtu, 
z tym, że adres, od którego rozpo-
czynany jest zapis danych powi-
nien być podzielny przez 32. Po 
nadaniu sekwencji START oraz 
adresu układu i adresu komórki 
pamięci układ nadrzędny transmi-
tuje do 32 bajtów przeznaczonych 
do zapisania w pamięci, które są 
potwierdzane sygnałem ACK, je-
śli wyprowadzenie /WC znajduje 
się w stanie niskim. Transmisja 
musi być oczywiście zakończona 
sygnałem STOP. Procedurę zapisu 
strony pamięci przedstawiono na 
list. 5.

Przed wywołaniem procedury, 
do zmiennej m24SrcAddr należy 
załadować adres początku bufo-
ra z danymi przeznaczonymi do 
zapisania, natomiast do zmiennej 
m24DstAddr  adres, pod który 
dane zostaną zapisane. W rejestrze 
A należy umieścić rozmiar zapisy-
wanej porcji danych (maksymalnie 
32 bajty). 

Ponieważ operacja programowa-
nia pamięci EEPROM może trwać 
nawet 5 milisekund, przed zaini-
cjowaniem kolejnej operacji zapisu 

danych należy od-
czekać na zakoń-
czenie poprzedniej 
operacji programo-
wania. Jeśli ope-
racja  programo-
wania nie została 
zakończona, przy 
p r ó b i e  z a a d r e -
sowania  układu 
M24C64 odpowie 

List. 2. Procedura zapisu danej do rejestru układu M41T56 
;***************************************************************************
; Procedura zapisu danej do rejestru układu M41T56
; Parametry:
;   A : dana do zapisu
;   X : adres rejestru
;***************************************************************************
.M41T56_WriteReg
 PUSH A          ; zapamiętanie danej na stosie
 PUSH X          ; zapamiętanie adresu rejestru na stosie
 CALL I2C_Start       ; START
 LD A, #%11010000       ; adres układu M41T56
 CALL I2C_Write       ; zapis adresu układu M41T56
 POP A          ; odtworzenie adresu rejestru ze stosu
 CALL I2C_Write       ; zapis adresu rejestru wewnętrznego
 POP A          ; odtworzenie danej do zapisu ze stosu
 CALL I2C_Write       ; zapis danej do rejestru wewnętrznego
 CALL I2C_Stop       ; STOP
 RET

List. 3. Procedura odczytu danych z rejestru układu M41T56 
;***************************************************************************
; Procedura odczytu zawartości rejestru układu M41T56
; Parametry:
;   X : adres rejestru
; Zwracana wartość :
;   A : odczytana dana
;***************************************************************************
.M41T56_ReadReg
 PUSH X          ; zapamiętanie adresu rejestru na stosie
 CALL I2C_Start       ; START
 LD A, #%11010000       ; adres układu M41T56
 CALL I2C_Write       ; zapis adresu układu M41T56
 POP A          ; odtworzenie adresu rejestru ze stosu
 CALL I2C_Write       ; zapis adresu rejestru wewnętrznego
 CALL I2C_Start       ; START
 LD A, #%11010001       ; adres układu M41T56 + Read
 CALL I2C_Write       ; zapis adresu układu M41T56 + Read
 SCF           ; NOACK
 CALL I2C_Read       ; odczyt zawartości rejestru układu M41T56
 CALL I2C_Stop       ; STOP
 RET

List. 4. Procedura zapisu pojedynczego bajtu danych do pamięci EEPROM 
;***************************************************************************
; Parametry :
;   m24DstAddr : adres komórki pamięci
;   A : dana do zapisu
;***************************************************************************
.M24C64_WriteByte
 PUSH A
 CALL I2C_Start       ; START
 LD A, #M24WRITEADDR     ;
 CALL I2C_Write       ; adres układu M24C64
 LD A, {m24DstAddr + 0}    ;
 CALL I2C_Write       ; MSB adresu komórki pamięci
 LD A, {m24DstAddr + 1}    ;
 CALL I2C_Write       ; LSB adresu komórki pamięci
 POP A
 CALL I2C_Write       ; dana do zapisania
 CALL I2C_Stop       ; STOP
 RET

List. 5. Procedura zapisu strony pamięci 
;***************************************************************************
; Parametry :
;   m24SrcAddr : adres bufora z danymi 
;   m24DstAddr : adres, pod który mają być zapisane dane
;   A : rozmiar bufora (max. 32 bajty)
;***************************************************************************
.M24C64_WriteBuffer
 LD m24Count, A       ; ilość bajtów do zapisu
 CALL I2C_Start       ; START
 LD A, #M24WRITEADDR     ;
 CALL I2C_Write       ; adres układu M24C64
 LD A, {m24DstAddr + 0}    ;
 CALL I2C_Write       ; MSB adresu komórki pamięci
 LD A, {m24DstAddr + 1}    ;
 CALL I2C_Write       ; LSB adresu komórki pamięci
 CLR X          ;
M24C64_WriteBufferLoop     ;
 LD A,([m24SrcAddr.w],X)    ; odczyt danych z bufora
 CALL I2C_Write       ; zapis danych do pamięci EEPROM
 INC X          ; X++
 CP X, m24Count       ; czy to ostatni bajt?
 JRNE M24C64_WriteBufferLoop  ; jeśli nie, to skok na początek pętli
 CALL I2C_Stop       ; STOP
 RET

Rys. 7. Przebiegi na magistrali I2C podczas odczytu danych z rejestrów zegara

Rys. 8. Struktura bajtu adresu EEPROM-u

Rys. 9. Przebiegi na magistrali I2C podczas zapisu pojedynczego bajtu do 
EEPROM-u

Rys. 10. Przebiegi na magistrali I2C podczas zapisu strony 32 bajtów do EEPROM-u

Rys. 11. Przebiegi na magistrali I2C podczas odczytu pojedynczego bajtu 
z EEPROMU-u

Rys. 12. Przebiegi na magistrali I2C podczas sekwencyjnego odczytu danych z EEPROM-u

on sygnałem NOACK. Dokonując 
cyklicznej próby zaadresowania 
układu M24C64 i sprawdzając stan 
bitu ACK możliwe jest stwierdze-
nie, czy operacja programowania 
pamięci dobiegła do końca. Proce-
durę realizującą to zadanie przed-
stawiono na list. 6. 

Odczyt danych z pamięci 
EEPROM

O d c z y t  d a n y c h  z  p a m i ę c i 
EEPROM, podobnie jak ich zapis 
może się odbywać na dwa spo-
soby: odczyt pojedynczego baj-
tu oraz odczyt 
strony. Przebie-
gi na magistrali 
podczas odczy-
tu pojedyncze-
g o  b a j t u  d a -
nych z pamięci 
przedstawiono 
na rys. 11. Pro-
cedurę odczy-
tu pojedyncze-
go bajtu spod dowolnego adresu 
przedstawiono na list. 7.

Przed wywołaniem procedury 
do zmiennej m24SrcAddr należy 
załadować adres komórki pamięci 
EEPROM, spod którego będą od-
czytywane dane. Odczytana dana 
zostanie umieszczona w rejestrze 
A.

W sytuacji, gdy 
musimy odczytać 
kolejno więcej niż 
jeden bajt, odczyty-
wanie pojedynczych 
baj tów może być 
bardzo uciążliwe. 
W związku  z  tym 
należy skorzystać 
z  m o ż l i w o ś c i  s e -
kwencyjnego odczytu danych z pa-
mięci M24C64. Przebiegi na magi-
strali podczas sekwencyjnego od-
czytu danych z pamięci EEPROM 
przedstawiono na rys. 12. Odczyt 
ostatniego bajtu danych musi zo-
stać zasygnalizowany bitem NO-
ACK, natomiast wszystkie poprzed-

nie bajty danych muszą zostać po-
twierdzone bitem ACK. Procedurę 
odczytującej dowolną ilość danych 
z pamięci EEPROM przedstawiono 
na list. 8. 

Przed wywołaniem procedu-
ry w zmiennej m24SrcAddr nale-
ży umieścić adres, spod którego 

będą odczytywane dane z pamięci 
EEPROM, natomiast w zmiennej 
m24DstAddr adres bufora w pa-
mięci RAM mikrokontrolera, do 
którego trafią odczytane dane.
W rejestrze A należy umieścić war-
tość określającą ilość odczytywa-
nych bajtów danych.

List. 6. Procedura sprawdzająca gotowość układu
;***************************************************************************
; Procedura oczekująca na bit ACK (zakończenie operacji programowania)
;***************************************************************************
.M24C64_WaitForACK
 CALL I2C_Start       ; START
 LD A, #M24WRITEADDR     ;
 CALL I2C_Write       ; Adres układu M24C64
 JRC M24C64_WaitForACK    ; Jeśli odebrano NOACK to skok na początek
 RET

List. 7. Procedura odczytu pojedynczego bajtu spod dowolnego adresu 
;***************************************************************************
; Parametry :
;   m24SrcAddr : adres, spod którego ma zostać odczytana dana
; Zwracana wartość :
;   A : odczytana dana
;***************************************************************************
.M24C64_ReadByte
 CALL I2C_Start       ; START
 LD A, #%10100000       ;
 CALL I2C_Write       ; adres układu slave
 LD A, {m24SrcAddr + 0}    ;
 CALL I2C_Write       ; MSB adresu komórki pamięci
 LD A, {m24SrcAddr + 1}    ;
 CALL I2C_Write       ; LSB adresu komórki pamięci
 CALL I2C_Start       ; START
 LD A, #%10100001       ;
 CALL I2C_Write       ; adres układu slave + R
 SCF           ; NOACK
 CALL I2C_Read       ; odczyt bajtu danych
 CALL I2C_Stop       ; STOP
 RET

List. 8. Procedura odczytu dowolnej ilości danych z pamięci EEPROM 
;***************************************************************************
; Parametry
;   m24SrcAddr : adres, spod którego mają być odczytane dane
;   m24DstAddr : adres bufora z danymi
;   A : rozmiar bufora (min. 2)
;***************************************************************************
.M24C64_ReadBuffer
 DEC A          ;
 LD m24Count, A       ; ilość bajtów do odczytu – 1
 CALL I2C_Start       ; START
 LD A, #%10100000       ;  
 CALL I2C_Write       ; adres układu slave
 LD A, {m24SrcAddr + 0}    ;
 CALL I2C_Write       ; MSB adresu komórki pamięci
 LD A, {m24SrcAddr + 1}    ;
 CALL I2C_Write       ; LSB adresu komórki pamięci
 CALL I2C_Start       ; START
 LD A, #%10100001       ;
 CALL I2C_Write       ; adres układu slave + R
 CLR X
M24C64_ReadBufferLoop
 RCF           ; ACK
 CALL I2C_Read       ; odczyt bajtu z pamięci EEPROM
 LD ([m24DstAddr.w],X),A    ; zapis do bufora
 INC X          ; X++
 CP X, m24Count       ; czy to przedostatni bajt?
 JRNE M24C64_ReadBufferLoop   ; jeśli nie to skok na początek pętli
 SCF           ; NOACK
 CALL I2C_Read       ; odczyt ostatniego bajtu z pamięci EEPROM
 LD ([m24DstAddr.w],X),A    ; zapis do bufora
 CALL I2C_Stop       ; STOP
 RET

Podsumowanie
Przedstawione w artykule proce-

dury są przeznaczone co prawda 
dla mikrokontrolerów ST7, jednakże 
mogą posłużyć za wzór przy pisaniu 
procedur obsługi układu M41T56C64 
dla innych mikrokontrolerów oraz ję-
zyków programowania. W przypadku 
tworzenia programu w języku dostar-
czającym gotowe procedury obsługi 
magistrali I2C, jak np. Bascom, czy 
też kompilatory mikroBasic/microPa-
scal/mikroC (http://www.mikroe.com/) 
wystarczy we właściwej kolejności 
wywołać podstawowe procedury od-

powiedzialne za transmisję sygna-
łów START, STOP oraz nadawanie 
i odbiór bajtu danych z magistrali. 
W przypadku tworzenia programu 
w asemblerze innej rodziny mikro-
kontrolerów należy dodatkowo opra-
cować własne procedury obsługi ma-
gistrali I2C, bądź też skorzystać z go-
towych procedur znalezionych np. 
w Internecie. Listingi przedstawionych 
w artykule procedur są zamieszczone 
na stronie internetowej Elektroniki 
Praktycznej oraz na płycie CD.

